
Housekeeping

⊚ This webinar is being recorded. A link to the recording will be sent out after the

presentation.

⊚ For questions on the content, please use the Zoom Q&A box (not the chat

panel). My colleagues will answer your questions there.

A Note on Sources

⊚ This presentation draws heavily on the Stata official manual, in particular

[BAYES] bayesmh and [BAYES] bayesmh evaluators , for both the methods

and the worked examples.

Quickstart

Stata’s Bayesian Suite

What the Bayesian module contains.

⊚ The bayesmh command for fitting arbitrary Bayesian models.

⊚ The bayes: prefix for fitting supported Bayesian models with familiar Stata estimation syntax.

⊚ A full set of Bayesian post-estimation commands ( bayesstats , bayestest , bayespredict )

and graphical diagnostics ( bayesgraph ).

How bayesmh and bayes: relate.

⊚ bayesmh is the foundation and core engine of Stata’s Bayesian framework.

⊚ Most bayes: prefix commands are built directly on bayesmh , supplying a convenience syntax

that assembles the underlying bayesmh call for you.

⊚ Both interfaces serve users at every experience level. bayes: is the faster entry point when your

model is on the supported list; bayesmh is where you go when you need the extra flexibility.

Quickstart

The MH Loop Inside bayesmh

bayesmh draws from the posterior with adaptive random-walk Metropolis–Hastings.

Each iteration runs the same four steps:

1 Propose. Draw a candidate parameter vector 𝜃∗ from a proposal distribution that

adaptation tunes during burn-in.

2 Score. Compute the log-posterior at the candidate: log 𝑝(𝑦 ∣ 𝜃∗) + log 𝑝(𝜃∗).
3 Decide. Accept 𝜃∗ with probability 𝛼 = min(1, 𝑝(𝜃∗∣𝑦)

𝑝(𝜃𝑡−1∣𝑦)); otherwise keep 𝜃𝑡−1.

4 Record. After burn-in, store each retained draw (respecting thinning) as the

posterior sample.

Quickstart

Built-in Likelihoods
Stata already ships a large catalog of likelihood families for bayesmh . Reach for the built-in first; only

drop to llf() or llevaluator() when your model is not on the list.

Where to find it: the [BAYES] bayesmh entry in the Stata PDF manual, under the likelihood() option.

Quickstart

Built-in Priors
Every declared parameter needs a prior. Stata ships a large library of univariate and multivariate prior

distributions that plug directly into prior() .

Quickstart

Built-in Priors (continued)

Where to find it: the [BAYES] bayesmh entry in the Stata PDF manual, under the prior() option. The flat ,

density() , and logdensity() entries at the bottom are the escape hatches when the catalog does not cover your

prior.

Quickstart

MCMC Controls: Definitions Only

Control bayesmh option What it does

Blocking block() Groups parameters into joint or separate update steps. Parameters in

the same block are proposed together.

Adaptation adaptation() Tunes the proposal distribution during burn-in to reach a target accep-

tance rate.

Thinning thinning() Retains every 𝑘th drawand, for fixed mcmcsize() , increases the total

MH iterations to collect the requested post-burn-in draws.

Burn-in burnin() Discards the first 𝑛 draws before storing.

These settings are out of scope for tuning here. For deterministic replication they still

must match across comparator runs.

Quickstart

Running bayesmh : A First Example
The simplest complete workflow on auto.dta : load, fit, and inspect.

sysuse auto, clear

quietly replace weight = weight/100

set seed 14

bayesmh mpg weight, ///

likelihood(normal({var})) ///

prior({mpg:}, flat) ///

prior({var}, jeffreys) ///

initial({mpg:weight} -0.6 {mpg:_cons} 39 ///

{var} 11.83)

⊚ likelihood(normal({var})) declares the observation model and exposes {var} as a

parameter you must cover with a prior.

⊚ prior() covers every declared parameter. Flat on the regression coefficients, Jeffreys on the

variance.

⊚ set seed plus explicit initial() pin the starting state so the run is reproducible.

Quickstart

Reading the Output

Bayesian normal regression MCMC iterations = 12,500

Random-walk Metropolis-Hastings sampling Burn-in = 2,500

MCMC sample size = 10,000

Number of obs = 74

Acceptance rate = .1668

Efficiency: avg = .04811

Log marginal-likelihood = -198.14302 [min, max rows omitted]

| Mean Std. dev. MCSE Median [95% cred. int.]

-------------+-------------------------------------------------------

mpg:weight | -.6026 .0541 .00267 -.6039 -.7115 -.5006

mpg:_cons | 39.505 1.6779 .0802 39.455 36.243 43.143

var | 12.266 2.1179 .0869 12.053 8.828 17.107

[values rounded; full-precision output in quickstart.log]

⊚ Header. Sampler, chain sizes, acceptance rate, and efficiency give a first read on how the chain

moved.

⊚ Posterior table. Per parameter: mean, SD, Monte Carlo standard error, median, and 95%

equal-tailed credible interval.

Quickstart

Convergence Diagnostics

bayesgraph diagnostics {mpg:weight}

Quickstart

Reading the Four Panels

Four panels, one question each. All four say the same thing on this run: no reason to doubt the posterior.

⊚ Trace. Is the chain stationary? Look for no drift and no stuck regions. Here: a flat band around

−0.60 across 10,000 iterations.

⊚ Histogram. Is the posterior unimodal? Here: one peak.

⊚ Density. Does it look the same in the first and second halves of the chain? Here: the All, 1-half,

and 2-half densities overlap.

⊚ Autocorrelation. Does it decay inside the default lag window? Here: near zero by lag ∼40, so

successive draws become effectively independent after a short gap.

Quickstart

bayesstats ess : Effective Sample Size

bayesstats ess

Efficiency summaries MCMC sample size = 10,000

Efficiency: min = .04114

avg = .04811

max = .05938

-----------------------------------------------------

| ESS Corr. time Efficiency

-------------+--------------------------------------

mpg |

weight | 411.36 24.31 0.0411

_cons | 438.19 22.82 0.0438

-------------+--------------------------------------

var | 593.75 16.84 0.0594

-----------------------------------------------------

⊚ ESS: effective number of independent draws the chain is worth. About 411 – 594 here, out of 10,000 kept

draws.

⊚ Corr. time: mcmcsize/ESS. Lag length needed for draws to be effectively independent; ∼17–24 lags on this

run.

⊚ Efficiency: ESS/mcmcsize. Fraction of kept draws carrying independent information. About 5% here.

Scope

Scope

What the rest of this talk covers.

⊚ Three ways to specify a likelihood in bayesmh : built-in likelihood() , inline llf() , and

user-written llevaluator() .

⊚ A matched-condition replication protocol for validating custom evaluator arithmetic against a

built-in comparator.

⊚ Worked examples: normal and logistic likelihoods on auto.dta in Stata 19.5.

Assumed background.

⊚ Basic Bayesian inference: prior, likelihood, posterior.

⊚ MCMC fundamentals: chains, burn-in, acceptance rate, Metropolis–Hastings.

⊚ Stata programming: user-written programs, local macros, and positional args binding.

Scope

Three Surfaces, One Rule

⊚ Surface 1: Built-in likelihood.

bayesmh mpg weight, likelihood(normal({var})) ...

⊚ Surface 2: Inline observation-level log-likelihood.

bayesmh mpg, ///

likelihood(llf( ///

lnnormalden(mpg, {mpg:weight}*weight + {mpg:_cons}, ///

sqrt({var})))) ...

⊚ Surface 3: Custom evaluator program.

bayesmh mpg weight, llevaluator(normalreg, parameters({var})) ...

Between Surface 2 and Surface 3 the syntax changes: you stop writing the linear

predictor yourself and start receiving it as xb .

Scope

Stop-Here Boundaries

Surface Stop here when What you write / verify

Built-in likelihood() Stata already supports the model Built-in command grammar; comparator

baseline; no custom-code audit

llf() Rowwise algebra still fits on one readable

line

One substitutable expression; comparator

script

llevaluator() Likelihood needs branching logic or extra

storage

Program block with observation-level log-

likelihood; comparator script

Writing Your Own Evaluator

llevaluator() vs. evaluator()

llevaluator() is a deliberately narrow interface. You write only the likelihood;

Stata handles priors, proposals, and acceptance.

llevaluator() evaluator()

You write Observation-level log-likelihood Log-likelihood and scalar lnprior

Signature args lnfj xb [params...] args lnfj lnprior xb [params...]

Stata still handles Priors, MH, adaptation, diagnostics MH, adaptation, diagnostics

prior() opt. Required Cannot be combined

Sampling Only Metropolis–Hastings Only Metropolis–Hastings

Best use Custom likelihood, standard priors Prior outside the built-in catalog and must be

coded

evaluator() is the exception case; llevaluator() remains preferred whenever

the prior fits prior() .

Writing Your Own Evaluator

Normal Evaluator: Define and Call

program normalreg

version 19.5

args lnfj xb var

tempname sd

scalar ‘sd’ = sqrt(‘var’)

quietly replace ‘lnfj’ = lnnormalden($MH_y, ‘xb’, ‘sd’) ///

if $MH_touse

end

bayesmh mpg weight, ///

llevaluator(normalreg, parameters({var})) ///

prior({mpg:}, flat) ///

prior({var}, jeffreys) ///

initial({mpg:weight} -0.6 {mpg:_cons} 39 {var} 11.83)

The whole picture : the evaluator program on top, the bayesmh call that invokes it below. The next two

slides walk through each block line by line.

Writing Your Own Evaluator

Normal Evaluator: The Full Program

program normalreg

version 19.5

args lnfj xb var

tempname sd

scalar ‘sd’ = sqrt(‘var’)

quietly replace ‘lnfj’ = lnnormalden($MH_y, ‘xb’, ‘sd’) ///

if $MH_touse

end

⊚ program normalreg : defines a Stata program bayesmh will call at every MCMC iteration.

⊚ version 19.5 : requests 19.5 syntax/behavior, guarding against future changes.

⊚ args lnfj xb var : positional binding (Slide 27).

⊚ tempname sd : temporary scalar name for intermediate storage.

⊚ scalar ‘sd’ = sqrt(‘var’) : variance to standard deviation. Getting this wrong is the bug on Slide 33.

⊚ quietly replace ‘lnfj’ = ... : the working line (Slide 29). Fills each in-sample row with the

observation-level log-likelihood.

Writing Your Own Evaluator

Calling the Evaluator

bayesmh mpg weight, ///

llevaluator(normalreg, parameters({var})) ///

prior({mpg:}, flat) ///

prior({var}, jeffreys) ///

initial({mpg:weight} -0.6 {mpg:_cons} 39 {var} 11.83)

Read these two blocks together. The first defines the callback. The second tells

bayesmh to call it.

bayesmh is the host: it allocates the workspace, passes in the inputs, and expects

your program to fill the output column correctly. Your evaluator is a guest that fills one

column and returns.

Writing Your Own Evaluator

One Evaluator Call, End to End

From the model specification bayesmh mpg weight , Stata parses the linear

predictor xb = {mpg:weight}*weight + {mpg:_cons} and recomputes it at every

iteration using the proposal’s current coefficient values.

Propose

params

(bayesmh)

Compute

xb

(bayesmh)

Call

evaluator

(your code)

Sum lnfj

+ prior

(bayesmh)

Accept /

Reject

(bayesmh)

Your evaluator has one job: compute the observation-level log-likelihood. It does not

propose, it does not accept or reject, and under llevaluator() it does not apply

priors.

Writing Your Own Evaluator

dryrun : Ask Stata What It Parsed
The dryrun script you run (appendix: preflight_dryrun.do ):

sysuse auto, clear

quietly replace weight = weight/100

bayesmh mpg weight, ///

likelihood(normal({var})) ///

prior({mpg:}, flat) ///

prior({var}, jeffreys) ///

initial({mpg:weight} -0.6 {mpg:_cons} 39 {var} 11.83) ///

dryrun

The parsed-model summary Stata prints back:

Likelihood:

mpg ~ normal(xb_mpg,{var})

Priors:

{mpg:weight _cons} ~ 1 (flat) (1)

{var} ~ jeffreys

(1) Parameters are elements of the linear form xb_mpg.

Writing Your Own Evaluator

What dryrun Catches

Once you move to llevaluator() , the visible summary stops at the callback name:

Likelihood:

mpg ~ normalreg(xb_mpg,{var})

Misspell {var} as {vvar} and dryrun catches it: return code 198 , no prior

specified for varname.

dryrun checks dryrun does not check

Likelihood syntax and declarations Evaluator algebra

Parameter declarations Custom prior algebra inside evaluator()

Prior coverage Posterior propriety

Writing Your Own Evaluator

Priors Are Part of the Model

bayesmh mpg weight, likelihood(normal({var})) ///

prior({mpg:_cons}, normal(25, {var})) ///

prior({mpg:weight}, normal(0, 100)) ///

prior({var}, igamma(5, 150))

Priors:

{mpg:_cons} ~ normal(25,{var})

{mpg:weight} ~ normal(0,100)

{var} ~ igamma(5,150)

⊚ Prefer prior specifications that dryrun can print back. Priors are part of the

model, so comparator runs must match priors exactly.

⊚ Trap: in prior(normal()) the second argument is variance, not standard

deviation. The same confusion shows up inside evaluator code on Slide 33.

© 2026 StataCorp LLC. All rights reserved. 26



Writing Your Own Evaluator

Argument Binding

bayesmh mpg weight, llevaluator(normalreg, parameters({var}))

What bayesmh provides to the evaluator (by position):

Slot 1: lnfj ← pre-allocated scratch column (fill with log-lik per obs)

Slot 2: xb ← linear predictor, recomputed each iteration

Slot 3: var ← first extra parameter from parameters({var})

program normalreg

args lnfj xb var

^ ^ ^

| | +-- binds to slot 3: first extra parameter

| +------- binds to slot 2: linear predictor

+------------- binds to slot 1: scratch column

parameters() binds by position, not by name. Like a patch bay: the first cable goes to the first port.

There is no runtime type check. If parameters() and args are out of order, the wrong value reaches

the wrong variable, silently.

Writing Your Own Evaluator

Ownership: What You May Read and Write

bayesmh hands your evaluator references into its workspace. Read its inputs. Write

only what the contract tells you to fill.

Target Owner Your evaluator should Common mistake

lnfj bayesmh alloc, you fill Overwrite every in-sample

row; cover every branch

generate on existing col-

umn; uncovered branches

xb bayesmh Read only Overwrite it

$MH_y bayesmh global Read the dependent-

variable name

Hardcode the outcome

$MH_touse bayesmh global Use as the sample mask on

every replace

Let off-sample rows con-

taminate the sum

Extras from parameters() bayesmh Read by position Assume name-based bind-

ing

Writing Your Own Evaluator

The Working Line and $MH_touse

quietly replace ‘lnfj’ = lnnormalden($MH_y, ‘xb’, ‘sd’) if $MH_touse

⊚ By contract, $MH_touse is mandatory.

⊚ replace , not generate . The scratch column already exists; generate would fail on iteration

two.

⊚ quietly suppresses per-call output. Without it, Stata prints a substitution message on every

evaluation.

Writing Your Own Evaluator

How the $MH_touse Guard Works

Row mpg $MH_touse Action under replace ... if $MH_touse

1 (in sample) 22 1 Overwrites lnfj with new log-likelihood

2 (missing data) . 0 Ignores row; lnfj keeps scratch value

3 (in sample) 17 1 Overwrites lnfj with new log-likelihood

Your job is to fill the in-sample cells, not to build the sheet from scratch.

Writing Your Own Evaluator

Correctness Bugs in Custom Evaluators

Bug Consequence

Pass variance to lnnormalden() instead of SD Wrong posterior: code runs, density is wrong, posterior shifts

Omit if $MH_touse Wrong posterior: off-sample rows contribute to the total log-

likelihood

Swap parameters() order Wrong posterior: wrong holder bound to wrong local macro; runs

silently

Leave a branch uncovered Wrong posterior: stale lnfj from previous proposal persists

Hardcode the outcome name Fragile: evaluator only works for one dataset layout

The first four mistakes silently shift the posterior; the last makes the evaluator

nonportable. Matched-condition comparison catches the first four when a built-in

comparator exists.

Writing Your Own Evaluator

Correct vs. Wrong
// CORRECT

scalar ‘sd’ = sqrt(‘var’)

quietly replace ‘lnfj’ = lnnormalden($MH_y, ‘xb’, ‘sd’) ///

if $MH_touse

// WRONG: passing variance, not standard deviation

quietly replace ‘lnfj’ = lnnormalden($MH_y, ‘xb’, ‘var’) ///

if $MH_touse

// WRONG: uncovered branch (only covers y == 1)

quietly replace ‘lnfj’ = ln(invlogit(‘xb’)) ///

if $MH_y == 1 & $MH_touse

// If y == 0, lnfj is unchanged and retains

// stale values from the previous iteration.

For evaluators with outcome-conditional branches, assert the data domain before

estimation:

assert inlist(foreign, 0, 1) if !missing(foreign, mpg)

Writing Your Own Evaluator

Catching the Canonical Bug

The bug is simple: pass variance instead of SD to lnnormalden() .

Matched-condition comparison against a deliberately broken evaluator:

Metric Built-in Broken evaluator Absolute diff

weight mean -0.6025616 -0.6020815 4.80e-4

_cons mean 39.50491 39.48684 1.81e-2

var mean 12.26586 3.47327 8.79259

Acceptance rate .1668 .2504 .08359

e(lml_lm) -198.14302 -198.93781 .79479

Same likelihood, priors, MCMC, seed, initials, data, Stata version, but different

posterior. 3.47327 is not random; it is near
√

12.26586 ≈ 3.50 because the broken

{var} is forced into the SD slot.

If you had compared only the regression coefficients, you might have missed the bug entirely.

Writing Your Own Evaluator

When Missing Values Lie

Boundary rejection Broken arithmetic

Aproposal stepsoutside the legal domain,

e.g. sqrt(var) with var < 0

The formula itself returns missing for a le-

gal proposal

Missing lnfj rejects the proposal Missing lnfj also rejects the proposal

Acceptable and expected Evaluator is broken

The sampler has no way to distinguish “the proposal landed outside the domain” from

“the formula has a bug.”

Writing Your Own Evaluator

Broken Logistic Evaluator

Broken evaluator used in the failure-mode demonstration:

program logitmiss

version 19.5

args lnfj xb

quietly replace ‘lnfj’ = . ///

if abs(‘xb’) > 2 & $MH_touse

quietly replace ‘lnfj’ = ln(invlogit(‘xb’)) ///

if abs(‘xb’) <= 2 & $MH_y == 1 & $MH_touse

quietly replace ‘lnfj’ = ln(invlogit(-‘xb’)) ///

if abs(‘xb’) <= 2 & $MH_y == 0 & $MH_touse

end

This is a failure-mode demonstration, not a comparator run. It uses mcmcsize(500) burnin(200) and

does not match the baseline MCMC settings. Its point: the log reaches a model summary instead of

stopping with an error, because missing lnfj is interpreted as rejection. Bad arithmetic can finish,

reject proposals, and still corrupt inference.

Matched-Condition Replication Protocol

Matched-Condition Replication Protocol

1 Baseline. Fit the model with the built-in likelihood under a fixed seed, explicit

initial values, and declared priors.

2 Test. Write the custom evaluator. Fit the same model with llevaluator()

under identicalmatched conditions: same likelihood, priors, MCMC settings,

parameterization, data, seed, initial values, and Stata version.

3 Extract. Store summaries from both runs: means, SDs, MCSEs, medians,

credible intervals, acceptance rate.

4 Compare. Compute absolute differences across all extracted metrics.

Matched-Condition Replication Protocol

Logistic Baseline

Built-in comparator:

set seed 14

bayesmh foreign mpg, likelihood(logit) ///

prior({foreign:}, flat) initial({foreign:} 0)

The normal family has an unknown variance that must be estimated, requiring

parameters({var}) . The binomial family has a fixed dispersion parameter of 1, so

all model uncertainty is captured by the regression coefficients in xb . The logistic

evaluator has no parameters() declaration; its signature is just args lnfj xb .

Three failure modes in double-precision arithmetic. Double-precision floating point can represent

values very close to 1 but not infinitely close. There are only finitely many representable values near 1,

and invlogit(37) crosses that boundary and rounds all the way to 1.0 .

Matched-Condition Replication Protocol

Three Arithmetic Failure Boundaries

Case Expression Failure mode Threshold

y = 0 ln(1 - invlogit(xb)) Cancellation after invlogit(xb)

rounds to 1

xb ≈ 37

y = 0 ln(invlogit(-xb)) exp(xb) overflows inside

invlogit(-xb)

xb ≈ 709.78

y = 1 ln(invlogit(xb)) exp(-xb) overflows;

invlogit(xb) = 0 ; ln(0) missing

xb ≈ -709.78

The manual’s shorter ln(invlogit()) form matches the built-in on auto.dta . For

datasets with more extreme predictors, the piecewise rewrite on the next slide avoids

all three failure boundaries.

Matched-Condition Replication Protocol

Safe Logistic Evaluator: Derivation

To avoid evaluating invlogit() directly, the log-probability decomposes into

branches. For y = 1 (i.e. log(sigmoid(xb)) ):

log(sigmoid(xb)) = -ln1p(exp(-xb)) for xb >= 0

log(sigmoid(xb)) = xb - ln1p(exp(xb)) for xb < 0

For y = 0 , the complementary identity 1 - sigmoid(xb) = sigmoid(-xb)

converts the problem to the same functional form, then expands identically:

log(sigmoid(-xb)) = -xb - ln1p(exp(-xb)) for xb >= 0

log(sigmoid(-xb)) = -ln1p(exp(xb)) for xb < 0

Matched-Condition Replication Protocol

logitllsafe

program logitllsafe

version 19.5

args lnfj xb

// y = 1 branches

quietly replace ‘lnfj’ = -ln1p(exp(-‘xb’)) ///

if $MH_y == 1 & ‘xb’ >= 0 & $MH_touse

quietly replace ‘lnfj’ = ‘xb’ - ln1p(exp(‘xb’)) ///

if $MH_y == 1 & ‘xb’ < 0 & $MH_touse

// y = 0 branches (via 1 - sigmoid(xb) = sigmoid(-xb))

quietly replace ‘lnfj’ = -‘xb’ - ln1p(exp(-‘xb’)) ///

if $MH_y == 0 & ‘xb’ >= 0 & $MH_touse

quietly replace ‘lnfj’ = -ln1p(exp(‘xb’)) ///

if $MH_y == 0 & ‘xb’ < 0 & $MH_touse

end

bayesmh foreign mpg, llevaluator(logitllsafe) ///

prior({foreign:}, flat) initial({foreign:} 0)

When invlogit() saturates, ln() of the saturated value is either trivially zero or undefined. The

piecewise rewrite avoids invlogit() entirely, so no saturation can occur.

Matched-Condition Replication Protocol

Logistic Validation Result

Both logistic evaluators match the built-in within machine precision. With MCSE

around 1.67e-3 on these parameters, the largest drift (∼ 1e-14 ) is about eleven
orders of magnitude smaller.

Evaluator Max summary diff* Acceptance-rate diff e(lml_lm) diff†

Manual ln(invlogit()) ∼3.6e-14 0 ∼7.1e-15

Piecewise ln1p() ∼2.4e-14 0 ∼1.4e-14

*Max absolute difference across mean, SD, MCSE, median, credible interval.

† Optional supplementary check.

Identical arithmetic paths give exact identity; rewritten paths give machine-precision

drift.

Local evidence on auto.dta (scope: Slide 16). This validates normal and logistic only. The next

evaluator you write still needs its own comparator run.

© 2026 StataCorp LLC. All rights reserved. 41



Matched-Condition Replication Protocol

Practice: The Blueprint Library ( [R] glm )

Your turn

Pick a GLM family from [R] glm Methods and Formulas and carry the sketch below through the

five-step protocol from Slide 36. The blueprint tells you which pieces stay fixed and which you have to

rewrite.

What stays fixed What changes by family

Evaluator contract: args lnfj xb ... , fill

lnfj , guard with $MH_touse

Inverse link and observation-level log-likelihood

formula

For families with ancillary parameters (gamma, inverse Gaussian, negative binomial), [R] glm gives

the likelihood kernel; you still must choose a Bayesian parameterization and prior for those extras in

bayesmh .

Starter sketch, Poisson with log link (not yet validated):

// For the log link, mu = exp(xb) substitutes directly.

// Same evaluator contract as normalreg.

quietly replace ‘lnfj’ = $MH_y*‘xb’ - exp(‘xb’) - ///

lngamma($MH_y + 1) if $MH_touse

No Matched-Condition Protocol

Weaker Checks (Part 1)

The protocol ends here. What follows are engineering heuristics for cases where

the protocol cannot be applied.

When no built-in comparator exists, work down a staged cross-check list. Treat a
novel evaluator like new lab hardware: calibrate the piece you can measure directly,
then add one moving part at a time.

1 Simplified subset. If your evaluator handles a complex model (e.g., zero-inflated Poisson),

validate the simpler component alone first against a built-in comparator (e.g., Poisson against

likelihood(poisson) ), then add one moving part at a time.

2 Fixed-parameter spot checks. Evaluate the log-likelihood at chosen parameter values and

compare with an independent calculation.

3 Simulated-data recovery.

B Calibration: generate repeated toy datasets from known parameters. Do credible

intervals cover the true values at the expected rate? (Stronger.)

B Sanity: on a single toy dataset, do posterior centers and intervals land near the

generating value? (Necessary, not sufficient.)

No Matched-Condition Protocol

Weaker Checks (Part 2)

4 Tiny-data hand calculation. Use a dataset small enough to compute the rowwise contributions

yourself.

5 Independent reimplementation. Recompute the same rowwise likelihood in another code path

and compare observation by observation.

When the comparator disappears, the evidence gets weaker. It does not become

optional.

Summary

Summary
⊚ Start with the smallest bayesmh surface that fits the model.

⊚ Run dryrun before every chain. It checks parsing and prior coverage, not evaluator algebra.

⊚ Read the evaluator as a contract: under llevaluator() , bayesmh owns proposals, xb , priors,

and acceptance; your code owns lnfj and the sample guard.

⊚ Positional mistakes are silent because there is no runtime type check.

⊚ Validate custom evaluator code by matched-condition replication against a built-in comparator

whenever one exists: same likelihood, priors, MCMC settings, parameterization, data, seed, initial

values, and Stata version.

⊚ Validation standard: exact zero when the arithmetic path is identical; example-justified

machine-precision tolerance when the algebra is rewritten. normalreg gives exact zero (same

lnnormalden() path); the logistic evaluator gives ∼ 1e-14 drift (rewritten ln1p() path).

Normal and logistic only.

⊚ If no comparator exists, say plainly that the evidence is weaker and work down the staged

cross-check list.

Before trusting a custom evaluator: run dryrun , match the model and sampler, fix seed

and initials, and compare against a built-in baseline.

